```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the approximation of the steady two dimensional rotational isentropic flow studied in page 436 of R. Courant, D. Hilbert, Methods of Mathematical Physics, Wiley Classics Library, Wiley, 1989,

$$
\left\{\begin{array}{c}
u \rho \frac{\partial \omega}{\partial x}+c^{2} \frac{\partial \sigma}{\partial x}=0  \tag{1}\\
u \rho \frac{\partial \lambda}{\partial x}+c^{2} \frac{\partial \sigma}{\partial y}=0 \\
\rho \frac{\partial \omega}{\partial x}+\rho \frac{\partial \lambda}{\partial y}+u \frac{\partial \sigma}{\partial x}=0
\end{array}\right.
$$

where $u$ denotes the constant velocity parallel to the $x$-axis, $\rho$ the constant density and $c$ the speed of sound. Let us introduce the ring $A=\mathbb{Q}(u, \rho, c)\left[d_{x}, d_{y}\right]$ of differential operators in $d_{x}$ and $d_{y}$ with coefficients in the field $\mathbb{Q}(u, \rho, c)$

```
> A:=DefineOreAlgebra(diff=[d[x],x], diff=[d[y],y], polynom=[x,y],
> comm=[u,rho,c]):
```

and the system matrix $R \in A^{3 \times 3}$ of (1) defined by:

```
> R:=matrix (3,3,[[u*rho*d[x], c^2*d[x],0],[0, c^2*d[y],u*rho*d[x]],
> [rho*d[x],u*d[x],rho*d[y]]]);
```

$$
R:=\left[\begin{array}{ccc}
u \rho d_{x} & c^{2} d_{x} & 0 \\
0 & c^{2} d_{y} & u \rho d_{x} \\
\rho d_{x} & u d_{x} & \rho d_{y}
\end{array}\right]
$$

We denote by $M=A^{1 \times 3} /\left(A^{1 \times 3} R\right)$ the $A$-module finitely presented by the matrix $R$. Let us study the endomorphism ring $E=\operatorname{end}_{A}(M)$ of $M$ :
> Endo:=MorphismsConstCoeff(R,R,A): Endo[1];

$$
\left[\left[\begin{array}{ccc}
1 & 0 & 0 \\
0 & 1 & 0 \\
0 & 0 & 1
\end{array}\right],\left[\begin{array}{ccc}
0 & c^{2} & 0 \\
0 & -u \rho & 0 \\
0 & 0 & -u \rho
\end{array}\right],\left[\begin{array}{ccc}
0 & 0 & c^{2} u \rho \\
0 & 0 & -u^{2} \rho^{2} \\
0 & -c^{2}\left(-c^{2}+u^{2}\right) & 0
\end{array}\right]\right]
$$

Hence, we obtain that $E$ is finitely generated by three endomorphisms $f_{1}=\operatorname{id}_{M}, f_{2}$ and $f_{3}$ defined by $f_{i}(\pi(\lambda))=\pi\left(\lambda P_{i}\right)$, where $\pi: A^{1 \times 3} \longrightarrow M$ denotes the projection onto $M, \lambda \in A^{1 \times 3}$ and $P_{i}$ is one of the three previous matrices. The generators $f_{i}$ 's of $E$ satisfy the relations Endo[2] ( $\left.\begin{array}{lll}f_{1} & f_{2} & f_{3}\end{array}\right)^{T}=0$, where Endo[2] is the following matrix:
$>$ Endo[2];

$$
\left[\begin{array}{ccc}
-u c^{2} \rho d_{x}+u^{3} d_{x} \rho & 0 & -d_{y} \\
0 & c^{2} d_{y} & d_{x} \\
0 & -c^{2} d_{x}+u^{2} d_{x} & d_{y}
\end{array}\right]
$$

Let us study the idempotents of the endomorphism ring $E$ defined by means of constant matrices, i.e., matrices defined by with zero-order differential operators:

```
> Idem:=IdempotentsConstCoeff(R,Endo[1],A,0,alpha);
```

$$
\begin{aligned}
& \text { Idem }:=\left[\left[\begin{array}{ccc}
0 & -1 / 2 \frac{c^{2}}{u \rho} & \alpha_{1} c \\
0 & 1 / 2 & -\frac{u \rho \alpha_{1}}{c} \\
0 & -\frac{\alpha_{1}\left(\left(-c^{2}+u^{2}\right)\right.}{u \rho} & 1 / 2
\end{array}\right],\left[\begin{array}{ccc}
1 & 0 & 0 \\
0 & 1 & 0 \\
0 & 0 & 1
\end{array}\right],\left[\begin{array}{ccc}
1 & \frac{c^{2}}{u \rho} & 0 \\
0 & 0 & 0 \\
0 & 0 & 0
\end{array}\right],\right. \\
& {\left.\left[\begin{array}{ccc}
0 & -\frac{c^{2}}{u \rho} & 0 \\
0 & 1 & 0 \\
0 & 0 & 1
\end{array}\right],\left[\begin{array}{ccc}
0 & 0 & 0 \\
0 & 0 & 0 \\
0 & 0 & 0
\end{array}\right],\left[\begin{array}{ccc}
1 & 1 / 2 \frac{c^{2}}{u \rho} & \alpha_{1} c \\
0 & 1 / 2 & -\frac{u \rho \alpha_{1}}{c} \\
0 & -\frac{\alpha_{1} c\left(-c^{2}+u^{2}\right)}{u \rho} & 1 / 2
\end{array}\right]\right] } \\
& {\left[\text { Ore_algebra, }\left[" d i f f^{\prime \prime}, " d i f f "\right],[x, y],\left[d_{x}, d_{y}\right],[x, y],\left[u, \rho, c, \alpha_{1}\right], 0,[],\right.} \\
& {\left.\left.\left[-1-4 \alpha_{1}^{2} c^{2}+4 \alpha_{1}^{2} u^{2}\right],[x, y],[],[],\left[\text { diff }=\left[d_{x}, x\right], \operatorname{diff}=\left[d_{y}, y\right]\right]\right]\right] }
\end{aligned}
$$

We obtain the two trivial idempotents 0 and $\operatorname{id}_{M}$ of $E$ respectively defined by the matrices 0 or $I_{3}$, two nontrivial idempotents respectively defined by the matrices $\operatorname{Idem}[1,3]$ and $\operatorname{Idem}[1,4]$ whose entries belong to $A$ and two non-trivial idempotents of $\operatorname{end}_{B}\left(B \otimes_{A} M\right)$, where $B=\mathbb{Q}(u, \rho, c)\left[\alpha_{1}\right] /\left(4\left(u^{2}-c^{2}\right) \alpha_{1}^{2}-1\right)\left[d_{x}, d_{y}\right]$, respectively defined by the matrices $\operatorname{Idem}[1,1]$ and $\operatorname{Idem}[1,6]$. Let us consider the matrix $P_{1}=\operatorname{Idem}[1,3]$ and $Q_{1}$ satisfying $R P_{1}=Q_{1} R$ :

$$
\begin{aligned}
&>P[1]:=\operatorname{Idem}[1,3] ; Q[1]:=\operatorname{Factorize}(\operatorname{Mult}(\mathrm{R}, \mathrm{P}, \mathrm{~A}), \mathrm{R}, \mathrm{~A}) ; \\
& P_{1}:=\left[\begin{array}{ccc}
1 & \frac{c^{2}}{u \rho} & 0 \\
0 & 0 & 0 \\
0 & 0 & 0
\end{array}\right] \quad Q_{1}:=\left[\begin{array}{ccc}
1 & 0 & 0 \\
0 & 0 & 0 \\
u^{-1} & 0 & 0
\end{array}\right]
\end{aligned}
$$

We can check that we have $P_{1}^{2}=P_{1}$ and $Q_{1}^{2}=Q_{1}$ :

```
> VERIF1:=simplify(evalm(Mult(P[1],P[1],A)-P[1]));
> VERIF2:=simplify(evalm(Mult(Q[1],Q[1],A)-Q[1]));
```

$$
\text { VERIF1 }:=\left[\begin{array}{ccc}
0 & 0 & 0 \\
0 & 0 & 0 \\
0 & 0 & 0
\end{array}\right] \quad \text { VERIF2 }:=\left[\begin{array}{lll}
0 & 0 & 0 \\
0 & 0 & 0 \\
0 & 0 & 0
\end{array}\right]
$$

Using the fact that matrices $P_{1}$ and $Q_{1}$ are idempotents of $A^{1 \times 3}$, we obtain that the $A$-modules $\operatorname{ker}_{A}\left(. P_{1}\right)$, $\operatorname{im}_{A}\left(. P_{1}\right)=\operatorname{ker}_{A}\left(.\left(I_{3}-P_{1}\right)\right), \operatorname{ker}_{A}\left(. Q_{1}\right)$ and $\operatorname{im}_{A}\left(. Q_{1}\right)=\operatorname{ker}_{A}\left(.\left(I_{3}-Q_{1}\right)\right)$ are projective, and thus, free by the Quillen-Suslin theorem. As the entries of $P_{1}$ and $Q_{1}$ only belong to the field $\mathbb{Q}(u, \rho, c)$, using linear algebraic techniques, we can easily compute bases of the corresponding $\mathbb{Q}(u, \rho, c)$-vector spaces, and thus, bases over the ring $A$ :

```
> U1:=SyzygyModule(P[1],A): U2:=SyzygyModule(evalm(1-P[1]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): V2:=SyzygyModule(evalm(1-Q[1]),A):
> V:=stackmatrix(V1,V2);
```

$$
U:=\left[\begin{array}{ccc}
0 & 1 & 0 \\
0 & 0 & 1 \\
u \rho & c^{2} & 0
\end{array}\right] \quad V:=\left[\begin{array}{ccc}
1 & 0 & -u \\
0 & 1 & 0 \\
1 & 0 & 0
\end{array}\right]
$$

The matrices $U \in \mathrm{GL}_{3}(A)$ and $V \in \mathrm{GL}_{3}(A)$ are such that the matrices $U P_{1} U^{-1}$ and $V Q_{1} V^{-1}$ are two block-diagonal matrices formed by the diagonal matrices $0_{2}$ and 1 :

```
> VERIF1:=Mult(U,P[1],LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q[1],LeftInverse(V,A),A);
```

$$
\text { VERIF1 }:=\left[\begin{array}{ccc}
0 & 0 & 0 \\
0 & 0 & 0 \\
0 & 0 & 1
\end{array}\right] \quad \text { VERIF2 }:=\left[\begin{array}{lll}
0 & 0 & 0 \\
0 & 0 & 0 \\
0 & 0 & 1
\end{array}\right]
$$

Then, the matrix $R$ is equivalent to the block-diagonal matrix $V R U^{-1}$ defined by:
> R_dec:=Mult(V,R,LeftInverse(U,A),A);

$$
R_{-} d e c:=\left[\begin{array}{ccc}
-d_{x}\left(-c^{2}+u^{2}\right) & -u \rho d_{y} & 0 \\
c^{2} d_{y} & u \rho d_{x} & 0 \\
0 & 0 & d_{x}
\end{array}\right]
$$

This last result can be obtained using the command HeuristicDecomposition:
> HeuristicDecomposition(R,P[1],A) [1];

$$
\left[\begin{array}{ccc}
-d_{x}\left(-c^{2}+u^{2}\right) & -u \rho d_{y} & 0 \\
c^{2} d_{y} & u \rho d_{x} & 0 \\
0 & 0 & d_{x}
\end{array}\right]
$$

Let us now consider the first $2 \times 2$ block-diagonal matrix $S$ of $R_{-} d e c$ defined by:
> S:=submatrix(R_dec,1..2,1..2);

$$
S:=\left[\begin{array}{cc}
-d_{x}\left(-c^{2}+u^{2}\right) & -u \rho d_{y} \\
c^{2} d_{y} & u \rho d_{x}
\end{array}\right]
$$

Let us try check whether or not the matrix $S$ is equivalent to a block-diagonal matrix. To do that, we introduce the $A$-module $L=A^{1 \times 2} /\left(A^{1 \times 2} S\right)$ finitely presented by the matrix $S$ and compute the endomorphism ring $F=\operatorname{end}_{A}(L)$ of $L$ :
> Endo1:=MorphismsConstCoeff(S,S,A): Endo1[1]; Endo1[2];

$$
\left[\left[\begin{array}{cc}
0 & u^{2} \rho^{2} \\
c^{2}\left(-c^{2}+u^{2}\right) & 0
\end{array}\right],\left[\begin{array}{ll}
1 & 0 \\
0 & 1
\end{array}\right]\right]\left[\begin{array}{cc}
d_{y} & -u c^{2} \rho d_{x}+u^{3} d_{x} \rho \\
d_{x} & u \rho c^{2} d_{y}
\end{array}\right]
$$

Let us check whether or not we can find idempotents of $F$ defined by means of constant matrices:

```
> Idem1:=IdempotentsConstCoeff(S,Endo1[1], A, 0, alpha);
```

$$
\text { Idem1 }:=\left[\left[\left[\begin{array}{cc}
1 / 2 & \frac{u \rho \alpha_{1}}{c} \\
\frac{\alpha_{1} c\left(-c^{2}+u^{2}\right)}{u \rho} & 1 / 2
\end{array}\right],\left[\begin{array}{ll}
0 & 0 \\
0 & 0
\end{array}\right],\left[\begin{array}{ll}
1 & 0 \\
0 & 1
\end{array}\right]\right],\right.
$$

[Ore_algebra, ["diff", "diff"], $[x, y],\left[d_{x}, d_{y}\right],[x, y],\left[u, \rho, c, \alpha_{1}\right], 0,[],\left[-1-4 \alpha_{1}^{2} c^{2}+4 \alpha_{1}^{2} u^{2}\right]$,

$$
\left.\left.[x, y],[],[],\left[\operatorname{diff}=\left[d_{x}, x\right], \text { diff }=\left[d_{y}, y\right]\right]\right]\right]
$$

We obtain the two trivial idempotents 0 and $\operatorname{id}_{F}$ of $F$ and an idempotent of $\operatorname{end}_{B}\left(B \otimes_{A} L\right)$, where $B=\mathbb{Q}(u, \rho, c)\left[\alpha_{1}\right] /\left(4\left(u^{2}-c^{2}\right) \alpha_{1}^{2}-1\right)\left[d_{x}, d_{y}\right]$, defined by the following matrix:

$$
\begin{aligned}
& >\mathrm{B}:=\operatorname{Idem} 1[2]: \mathrm{P}[2]:=\operatorname{Idem} 1[1,1] ; \mathrm{Q}[2]:=\operatorname{Factorize}(\operatorname{Mult}(\mathrm{S}, \mathrm{P}[2], \mathrm{B}), \mathrm{S}, \mathrm{~B}) ; \\
& P_{2}:=\left[\begin{array}{cc}
1 / 2 & \frac{u \rho \alpha_{1}}{c} \\
\frac{\alpha_{1} c\left(-c^{2}+u^{2}\right)}{u \rho} & 1 / 2
\end{array}\right] \quad Q_{2}:=\left[\begin{array}{cc}
1 / 2 & \frac{\alpha_{1} c^{2}-\alpha_{1} u^{2}}{c} \\
-\alpha_{1} c & 1 / 2
\end{array}\right]
\end{aligned}
$$

We can check that the matrices $P_{2}$ and $Q_{2}$ satisfy $P_{2}^{2}=P_{2}$ and $Q_{2}^{2}=Q_{2}$ :

```
> VERIF1:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),alpha[1]^4=1/16/(u^2-c^2)^2,
> simplify(evalm(Mult(P[2],P[2],B)-P[2]))));
> VERIF2:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),alpha[1]^4=1/16/(u^2-c^2)^2,
> simplify(evalm(Mult(Q[2],Q[2],B)-Q[2]))));
\[
\text { VERIF1 }:=\left[\begin{array}{ll}
0 & 0 \\
0 & 0
\end{array}\right] \quad \text { VERIF2 }:=\left[\begin{array}{ll}
0 & 0 \\
0 & 0
\end{array}\right]
\]
```

As the matrices $P_{2}$ and $Q_{2}$ are idempotents of $B^{2 \times 2}$, we know that the $B$-modules $\operatorname{ker}_{B}\left(. P_{2}\right), \operatorname{im}_{B}\left(. P_{2}\right)=$ $\operatorname{ker}_{B}\left(.\left(I_{2}-P_{2}\right)\right), \operatorname{ker}_{B}\left(. Q_{2}\right)$ and $\operatorname{im}_{B}\left(. Q_{2}\right)=\operatorname{ker}_{B}\left(.\left(I_{2}-Q_{2}\right)\right)$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of those free $B$-modules:

```
> X1:=SyzygyModule(P[2],B):X2:=SyzygyModule(evalm(1-P[2]),B):
> X:=stackmatrix(X1,X2);
> Y1:=SyzygyModule(Q[2],B): Y2:=SyzygyModule(evalm(1-Q[2]),B):
> Y:=stackmatrix(Y1,Y2);
\[
X:=\left[\begin{array}{cc}
-2 \alpha_{1} c u^{2}+2 \alpha_{1} c^{3} & u \rho \\
-2 \alpha_{1} c^{3}+2 \alpha_{1} c u^{2} & u \rho
\end{array}\right] \quad Y:=\left[\begin{array}{cc}
-2 \alpha_{1} c & -1 \\
-2 \alpha_{1} c & 1
\end{array}\right]
\]
```

We can easily check that $X P_{2} X^{-1}$ and $Y Q_{2} Y^{-1}$ are the block-diagonal matrices $\operatorname{diag}(0,1)$ :

```
> VERIF1:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),Mult(X,P[2],
> LeftInverse(X,B),B)));
> VERIF2:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),Mult(Y,Q[2],
> LeftInverse(Y,B),B)));
\[
\text { VERIF1 }:=\left[\begin{array}{ll}
0 & 0 \\
0 & 1
\end{array}\right] \quad \text { VERIF2 }:=\left[\begin{array}{ll}
0 & 0 \\
0 & 1
\end{array}\right]
\]
```

Then, the matrix $S$ is equivalent to the block-diagonal matrix $Y S X^{-1}$ defined by:

$$
\begin{aligned}
& >\text { S_dec: }=\text { simplify }\left(\operatorname{subs}\left(\operatorname{alpha}[1]^{\wedge} 2=1 / 4 /\left(u^{\wedge} 2-c^{\wedge} 2\right), \text { Mult }(Y, S, L e f t I n v e r s e(X, B), B)\right)\right) ; \\
& \qquad S_{-} d e c:=\left[\begin{array}{cc}
\frac{c d_{y}+2 d_{x} \alpha_{1} c^{2}-2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)} & 0 \\
0 & \frac{c d_{y}-2 d_{x} \alpha_{1} c^{2}+2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)}
\end{array}\right]
\end{aligned}
$$

This last result can be directly obtained as follows:

```
> simplify(subs(alpha[1]^2=1/4/(u^2-c^2),HeuristicDecomposition(S,P[2],B)
> [1]));
```

$$
\left[\begin{array}{cc}
\frac{c d_{y}+2 d_{x} \alpha_{1} c^{2}-2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)} & 0 \\
0 & \frac{c d_{y}-2 d_{x} \alpha_{1} c^{2}+2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)}
\end{array}\right]
$$

If we denote by

$$
\begin{aligned}
& >\mathrm{G}:=\operatorname{diag}(\mathrm{X}, 1): \mathrm{H}:=\operatorname{diag}(\mathrm{Y}, 1): \mathrm{Z}:=\operatorname{Mult}(\mathrm{G}, \mathrm{U}, \mathrm{~B}) ; \mathrm{T}:=\operatorname{Mult}(\mathrm{H}, \mathrm{~V}, \mathrm{~B}) ; \\
& Z:=\left[\begin{array}{ccc}
0 & -2 \alpha_{1} c u^{2}+2 \alpha_{1} c^{3} & u \rho \\
0 & -2 \alpha_{1} c^{3}+2 \alpha_{1} c u^{2} & u \rho \\
u \rho & c^{2} & 0
\end{array}\right] \quad T:=\left[\begin{array}{ccc}
-2 \alpha_{1} c & -1 & 2 \alpha_{1} c u \\
-2 \alpha_{1} c & 1 & 2 \alpha_{1} c u \\
1 & 0 & 0
\end{array}\right]
\end{aligned}
$$

then the matrix $R$ is equivalent to the simple block-diagonal matrix $T R Z^{-1}$ defined by:

```
> simplify(subs(alpha[1] ~2=1/4/(u^2-c^2),simplify(Mult(T,R,LeftInverse(Z,B),
> B))));
```

$$
\left[\begin{array}{ccc}
\frac{c d_{y}+2 d_{x} \alpha_{1} c^{2}-2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)} & 0 & 0 \\
0 & \frac{c d_{y}-2 d_{x} \alpha_{1} c^{2}+2 d_{x} \alpha_{1} u^{2}}{2 \alpha_{1}\left(u^{2}-c^{2}\right)} & 0 \\
0 & 0 & d_{x}
\end{array}\right]
$$

If $\mathcal{F}$ denotes an $A$-module (e.g., $\mathcal{F}=C^{\infty}\left(\mathbb{R}^{2}\right)$ ), using the relation $2 \alpha_{1}\left(c^{2}-u^{2}\right)=-1 /\left(2 \alpha_{1}\right)$, we then obtain that the linear system $\operatorname{ker}_{\mathcal{F}}(R$.$) is equivalent to the following one$

$$
\left\{\begin{array} { l } 
{ ( d _ { x } - 2 c \alpha _ { 1 } d _ { y } ) \zeta _ { 1 } = 0 , } \\
{ ( d _ { x } + 2 c \alpha _ { 1 } d _ { y } ) \zeta _ { 2 } = 0 , } \\
{ d _ { x } \zeta _ { 3 } = 0 }
\end{array} \Leftrightarrow \left\{\begin{array}{l}
\zeta_{1}=\phi\left(y+2 c \alpha_{1} x\right) \\
\zeta_{2}=\psi\left(y-2 c \alpha_{1} x\right) \\
\zeta_{3}=C
\end{array}\right.\right.
$$

where $\phi$ and $\psi$ are two arbitrary functions of $\mathcal{F}$ and $C$ an arbitrary constant.

